EE 236: Experiment No. 2
I-V Characteristics of Schottky and Zener Diodes

Vinamra Baghel, 190010070

January 23, 2022

1 Overview of the experiment

1.1 Aim of the experiment

The following were the aims of the experiment:

- To plot I-V characteristics of Schottky diodes and extract parameters (like reverse recovery time) and to compare them with those of a PN Junction diode.
- To plot I-V characteristics of Zener diode and find its usage in a Zener regulator.
- To design and simulate circuits for given transfer characteristics.
- To simulate a Voltage Doubler circuit.

1.2 Methods

1.2.1 Schottky Diode

We started by writing netlists for BAT85, BAT960, and 1N914 PN junction diodes, according to the following circuit diagram given in Lab 1, in a single CIR file for plotting and comparing the three I-V characteristics. DC analysis was performed on the diodes to obtain the characteristics. The cut-in voltages were noted from the graph.
The PN junction diodes in the bridge rectifier were replaced by the Schottky diodes and outputs were obtained.
Next, reverse recovery times of the diodes were compared using a unit pulse input of 100 kHz.

1.2.2 Zener Diode

The I-V characteristics were plotted by writing the netlist according to the circuit given in Lab 1. DC analysis was performed from -20 to 10 volts.
Next task was to simulate a Zener regulator circuit (given below). Output from the bridge rectifier with BAT960 diode and a capacitive filter was taken as the unregulated 15V input. The values of R_S, R_1, and R_2 were obtained using given parameters. The plot of the regulated 9V output was obtained.

1.2.3 Circuit Design from given Transfer Characteristics

We had to figure out the circuits for the given transfer characteristics. After that, we plotted the graphs using obtained netlists and confirmed our hypotheses.

1.2.4 Voltage Doubler

The netlist was written according to the given circuit diagram. 1N914 diode and a 20V 1kHz sinusoidal input were used for the voltage doubler. Finally the plots were obtained.

2 Design

2.1 Schottky Diode

The netlist included the three diodes and corresponding 1 kΩ resistors, and an input voltage source. We had three parallel circuits with a common voltage source and ground. DC analysis was performed by varying V_{in} from 0.01 to 5V in steps of 0.01V to obtain the I-V characteristics. The circuit diagram is given below.
We plotted I vs V across the diode which is given by:

$$I_D = I_{00} e^{-\frac{E_g}{kT}} (e^{\frac{qV_D}{kT}} - 1)$$



Figure 1: The IV Characterization Circuit

Next part was to obtain reverse recovery times of the three diodes. This was done by plotting the graphs for simple circuits and obtaining the results. The diodes used for comparison were BAT85, BAT960 and 1N4007.

2.2 Zener Diode

I-V characteristics were obtained in the same way as described above.
Next part was to simulate a Voltage Regulator circuit. Output from a **capacitive filtered bridge rectifier**, whose input was a 20 V 50 Hz sinusoid, was used as input and the following circuit was made:



Figure 2: Zener Voltage Regulator

The values of the resistor R_S was obtained as 52.64 Ω using the diode wattage and breakdown voltage according to the following equations:

$$I_Z.R_S = V_S - V_Z$$

$$I_Z.V_Z = 1$$

The values of the resistors R_1 and R_2 were obtained as 6 kΩ and 10 kΩ respectively using the expected output voltage according to the following equation:

$$R_1 = 6k\Omega$$

$$\frac{5.6}{R_2} = \frac{9 - 5.6}{R_1}$$

The circuit was simulated and plot for the output voltage waveform obtained.

2.3 Circuit Design from given Transfer Characteristics

The first plot seemed to be similar to a shunt clipper from Lab 0. It had an input offset of 0.7 V and 1.6 V of an external current source. The hypothesis was verified by obtaining plots using NgSpice.



Figure 3: First Circuit from Transfer Characteristics

The second plot seemed to be a mirrored version of the shunt clipper. It had to have another similar branch acting for the negative side. The hypothesis was verified by obtaining plots using NgSpice.



Figure 4: Second Circuit from Transfer Characteristics

2.4 Voltage Doubler

The given circuit for the voltage doubler was simulated. The netlist included two 10 μF capacitors, two 1N914 diodes, and a 4.7 (/47) kΩ resistor along with a 20 V 1 kHz sinusoidal input. Plot for the output waveform was obtained.



Figure 5: Voltage Doubler

3 Simulation results

3.1 Code snippets

3.1.1 Schottky Diode

```
Vinamra Baghel 190010070 Schottky IV Characteristics
.include schottky_BAT85.txt
.include schottky_BAT960.txt
.include Diode_1N914.txt

*Netlist
r1 in 1 100
r2 3 gnd 100
```

```
r3 1 gnd 1k
X1 1 2 BAT85
Vdr1 2 3 0

r4 in 4 100
r5 6 gnd 100
r6 4 gnd 1k
X2 4 5 BAT960
Vdr2 5 6 0

r7 in 7 100
r8 9 gnd 100
r9 7 gnd 1k
D3 7 8 1N914
Vdr3 8 9 0

Vin in gnd 0

*Analysis
.dc Vin 0 5 0.01
.control
run
plot I(Vdr1) vs V(1)-V(2) I(Vdr2) vs V(4)-V(5) I(Vdr3) vs V(7)-V(8)
.endc
.end
```

```
Vinamra Baghel 190010070 Schottky Bridge Rectifier
.include schottky_BAT85.txt
.include schottky_BAT960.txt

*Netlist
rl out1 out2 1k
X1 in out2 BAT85
X2 out1 in BAT85
X3 out1 gnd BAT85
X4 gnd out2 BAT85
Vin in gnd sin(0 12 50 0 0 0)

*Analysis
.tran 1u 60m
.control
run
plot V(out2) - V(out1)
.endc
.end
```

```
Vinamra Baghel 190010070 Reverse Recovery
.include schottky_BAT85.txt
.include schottky_BAT960.txt
.include 1N4007.txt
```

```

*Netlist
r85 in 1 1k
X85 1 2 BAT85
Vd1 2 gnd 0

r960 in 3 1k
X960 3 4 BAT960
Vd2 4 gnd 0

r in 5 1k
D 5 6 DI_1N4007
Vd3 6 gnd 0

Vin in gnd pulse(-1 1 0 0 0 5u 10u)

*Analysis
.tran 0.1u 10u
.control
run
plot I(Vd1) I(Vd2) I(Vd3)
.endc
.end
```

3.1.2 Zener Diode

```
Vinamra Baghel 190010070 Zener IV Characteristics
.include zener.txt

*Netlist
r1 in 1 100
r2 3 gnd 100
r3 1 gnd 1k
X 1 2 DI_1N4734A
Vdr 2 3 0
Vin in gnd 0

*Analysis
.dc Vin -20 10 0.01
.control
run
plot I(Vdr) vs V(1)-V(2)
.endc
.end
```

```
Vinamra Baghel 190010070 Voltage Regulator
.include schottky_BAT960.txt
.include bc547.txt
.include zener.txt
.include ua741.txt
```

```

*Netlist
rl out1 out2 1k
X1 in out2 BAT960
X2 out1 in BAT960
X3 out1 gnd BAT960
X4 gnd out2 BAT960
rl1 out2 out1 1k
cl out2 out1 500u
Vin in gnd sin(0 20 50 0 0 0)

rs out2 p 52.64
r1 out n 6k
r2 n out1 10k
rl2 out out1 1k
Xz out1 p DI_1N4734A
Q out2 b out bc547a
Xo p n pp nn b ua741
Vpp pp gnd 12
Vnn nn gnd -12

*Analysis
.tran 1u 50m
.control
run
plot V(out)-V(out1)
.endc
.end
```

3.1.3 Circuit Design from given Transfer Characteristics

```
Vinamra Baghel 190010070 TC1
.include Diode_1N914.txt

*Netlist
r 1 2 1k
D 2 3 1N914
V7 1 in 0.7
V16 3 gnd 1.6
Vin in gnd 0

*Analysis
.dc Vin -5 5 0.01
.control
run
plot V(2)
.endc
.end
```

```
Vinamra Baghel 190010070 TC2
```

```
.include Diode_1N914.txt

*Netlist
r in 1 1k
D1 1 2 1N914
D2 3 1 1N914
V48 2 gnd 4.8
V40 gnd 3 4
Vin in gnd 0

*Analysis
.dc Vin -6 6 0.01
.control
run
plot V(1)
.endc
.end
```

3.1.4 Voltage Doubler

```
Vinamra Baghel 190010070 Voltage Doubler
.include Diode_1N914.txt

*Netlist
r out gnd 47k
c1 in 1 10u
c2 out gnd 10u
D1 gnd 1 1N914
D2 1 out 1N914
Vin in gnd sin(0 20 1k 0 0 0)

*Analysis
.tran 1u 20m
.control
run
plot V(out)
.endc
.end
```

3.2 Simulation results

3.2.1 Schottky Diode



Figure 6: Schottky and PN Junction Diode I-V Characteristics

Index:

- **Red:** BAT85 Schottky
- **Blue:** BAT960 Schottky
- **Green:** 1N914 PN Junction



Figure 7: Bridge Rectifier using BAT85 Schottky Diode



Figure 8: Reverse Recovery Time Comparison

Index:

- **Blue:** BAT85 Schottky
- **Red:** BAT960 Schottky
- **Brown:** 1N4007 PN Junction

3.2.2 Zener Diode



Figure 9: Zener I-V Characteristics



Figure 10: Voltage Regulator Output

3.2.3 Circuit Design from given Transfer Characteristics



(a) First TC Circuit

(b) Second TC Circuit

Figure 11: Transfer Characteristics (TC) Circuits

3.2.4 Voltage Doubler



Figure 12: E_g vs V_d

4 Experimental results

The following cut-in voltage values were obtained for the Schottky diodes, 1N914, and Zener diode from the respective I-V Characteristics plots.

Diode	Cut-in Voltage (in mV)
BAT85 Schottky	160
BAT960 Schottky	110
1N914	700
Zener	450

The breakdown voltage obtained for the Zener diode was **5.4 V**
. The following reverse recovery timings were obtained for the Schottky diodes and 1N4007 from the respective Reverse Recovery plots.

Diode	Reverse recovery time (in μs)
BAT85 Schottky	0.05
BAT960 Schottky	0.2
1N4007	0.9

5 Experiment completion status

I could complete all the parts of the lab. There was no hardware involvement as it was all simulation based. The results were shown to the TA and then submitted.

6 Questions for reflection

Question 1. What are the limiting factors those will affect the limit of the load? How will you overcome those?
The maximum OpAmp output current is limited. Due to this, the load cannot be too small or the OpAmp might get damaged. One solution certainly is to use the BJT. Another would be to use a better OpAmp with greater current output.

Question 2. Why the BJT is required here?
The BJT boosts the output current by the OpAmp which is extremely small. This lets us have a lower load resistance.

Question 3. What will be the change in output if the resistor is replaced by a 4.7 kΩ resistor?
With a 47 kΩ resistor, the average output voltage is 38.2 V. With a 4.7 kΩ resistor, the average output voltage drops to 37.3 V. Also, with the 4.7 kΩ resistor, the output oscillates more compared to with the 47 kΩ resistor.

Question 4. What are the voltages to which the capacitors C1 and C2 in the circuit charge?
C1 charges to 20 V. C2 charges to the output voltage, 38 V, i.e. almost 40 V.

Question 5. Comment about few applications where schottky diode can be used.
In power electronics, where fast switching circuits are involved.